```
% ********************
% Parties d'un ensemble
% ********************
% ex : ?- partie([1,2,3],  P).

partie([],[]).
partie([X|L],[X|P]) :-
        partie(L,P).
partie([_X|L],P) :-
        partie(L,P).


% *************************
% Base de donnees de graphes
% *************************
% on modelise un graphe par une relation prolog du type : graphe(Nom, Liste_Sommets, Liste_Arcs)

graphe(g1, [1,2,3,4,5,6], [[1,2], [1,3], [2,4], [3,4], [3,5], [4,6], [5,6]]).
graphe(g2, [1,2,3,4,5,6], [[1,2], [1,3], [2,4], [3,4], [3,5], [4,1], [4,6], [5,6],[6,3]]).

% ********************************************************
% Relations entre sommet et graphe,  et entre arc et graphe
% ********************************************************

sommet(G,X) :- graphe(G,Sommets,_A), member(X,Sommets).

arc(G,O,D) :- graphe(G,_S,Arcs), member([O,D],Arcs).

% **********************************
% Existence et construction de chemins
% **********************************

existe_chemin(G,O,D) :-
        arc(G,O,D).
existe_chemin(G,O,D) :-
        arc(G,O,I),
        existe_chemin(G,I,D).

chemin(G,O,D,[[O,D]]) :-
        arc(G,O,D).
chemin(G,O,D,[[O,I]|C]) :-
        arc(G,O,I),
        chemin(G,I,D,C).

% ************************************
% Cas de graphes comportant des circuits
% ************************************

% Il faut eviter d'atteindre deux fois le meme sommet
% Pour cela on utilise un argument supplémentaire, Memo,
% qui accumule les sommets déja quittes. Pour eviter de boucler
% il faut utiliser des arcs qui n'ont pas de destination
% contenue dans Memo.
% A l'appel, Memo initial est fourni ; c'est une liste vide
% puisqu'on n'a encore pas emprunté d'arc.

% On cache l'utilisation de cet argument supplémentaire en
% proposant plutot un predicat à 4 arguments, chemin2

chemin2(G,O,D,C) :- chemin_ss_circuit(G,O,D,[],C).

chemin_ss_circuit(G,O,D,Memo,[[O,D]]) :-
        arc(G,O,D),
        not member(D,Memo).
chemin_ss_circuit(G,O,D,Memo,[[O,I]|C]) :-
        arc(G,O,I),
        not member(I,Memo),
        chemin_ss_circuit(G,I,D,[O|Memo],C).
```

```
% exemple d'appel : ?-   chemin2(g2, X,Y,Chemin)
```